

Anhang C: Recodierung

Die nachfolgend aufgeführte Syntax wurde in SPSS erstellt, um die Daten auf ein gleiches Skalenniveau umzuberechnen. Dazu wurde für jede Qualitätsdimension die jeweils zugehörigen Variablen auf einen Wert zwischen 1 und 10 umgerechnet und anschliessend zu einem Index zusammengefasst. Die fett gedruckten Zeilen sind Kommentare und wurden nachträglich zur besseren Orientierung in die Syntax eingefügt.

Aktualität

Differenz zwischen Ereigniszeitpunkt und Publikationszeitpunkt

```
COMPUTE zeitdif = DATEDIFF (V3, V10, "days").
```

```
IF (zeitdif <= 7 AND zeitdif >= 0) publ_geschw = 10 - (zeitdif * (10/7)).
```

```
IF zeitdif > 7 publ_geschw = 0.
```

```
IF zeitdif < 0 publ_geschw = 10.
```

```
IF (MISSING(zeitdif) = 1 AND V11 = 1) publ_geschw = 5.
```

```
IF V11 = 0 publ_geschw = 0.
```

Nachträgliche Aktualisierungen

```
IF V45 >= 1 aktual = 10.
```

```
IF V45 = 0 aktual = 0.
```

```
IF MISSING(V45) = 1 aktual = 0.
```

Bezug zu älteren Artikeln

```
RECODE V13 (0=0) (1=10) INTO rueck.
```

```
IF MISSING(V13) = 1 rueck = 0.
```

Beantwortung von Leser-Kommentaren

```
IF (V102 = 1 AND V103 = 1) beantw = 10.
```

Publikationshäufigkeit auf Angebotsebene

```
RECODE V2
```

```
(1 = 10) (2 = 2.66) (3 = 2.9408) (4 = 0.7492) (5 = 0.7316) (6 = 4.3318)
```

```
(7 = 3.9933) (8 = 9.8139) (9 = 4.4248) (10 = 2.8606) (11 = 3.2087)
```

```
(12 = 8.0218) (13 = 4.8131)
```

```
INTO publ_haeuf.
```

Berechnung des Index für Aktualität

```
COMPUTE aktualitaet = MEAN(publ_geschw, aktual, rueck, beantw, publ_haeuf).
```

Transparenz

Ersichtlichkeit des Autors

```
COMPUTE autor_ers = V44 * (10/3).
```

Ersichtlichkeit der Zeitpunkte von Nachträgen und der Publikation

```
RECODE V12 (0 = 0) (1 = 2) (2 = 1) INTO zeitp_pub.  
COMPUTE zeitp_pub = (zeitp_pub * (10 / 2)).
```

```
RECODE V14 (0 = SYSMIS) (1, 2 = 10) INTO zeitp_nac.
```

```
COMPUTE zeitpkt = MEAN(zeitp_pub, zeitp_nac).
```

Ersichtlichkeit der Quellen

```
COMPUTE quel_anz = V25 + V26 + V27 + V28 + V29 + V31 + V32 + V34.
```

```
DO IF quel_anz > 0.  
COMPUTE ers_urh = ((10 / quel_anz) * V38).  
COMPUTE ers_ort = ((10 / quel_anz) * V39).  
COMPUTE ers_zeit = ((10 / quel_anz) * V40).  
COMPUTE ers_link = ((10 / quel_anz) * V41).  
COMPUTE quel_ers = MEAN(ers_urh, ers_ort, ers_zeit, ers_link).  
END IF.
```

Quellentransparenz: Links

```
DO IF V20 > 0.  
COMPUTE links_ziel = ((10 / V20) * (V20 - V23)).  
COMPUTE links_deep = ((10 / V20) * V24).  
END IF.
```

```
COMPUTE links_trans = MEAN(links_ziel, links_deep).
```

Methoden genannt

```
RECODE V42 (0 = 0) (1 = 10) INTO method.
```

Quellenkontexte genannt

```
RECODE V43 (0 = 0) (1 = 10) INTO quell_kont.
```

Korrigenda und korrektiver Bezug

```
RECODE V46 (0 = 0) (1 = 10) INTO korrig.  
RECODE V47 (0 = 0) (1 = 10) INTO korr_bez.
```

Trennung von Kommentar und Meinung

```
RECODE V53 (0 = 0) (1 = 10) INTO meing_tren.
```

Beschriftung der Illustrationen

```
NUMERIC ill_beschr_20 ill_beschr_21 ill_beschr_22 ill_beschr_23 ill_beschr_24.  
COMPUTE anz_ill_typ = 0.
```

```
RECODE V120d (SYSMIS = 0).  
RECODE V121d (SYSMIS = 0).  
RECODE V122d (SYSMIS = 0).  
RECODE V123d (SYSMIS = 0).  
RECODE V124d (SYSMIS = 0).
```

```
IF (V120a > 0 OR V120b > 0) ill_beschr_20 = V120d * 5.  
IF (V121a > 0 OR V121b > 0) ill_beschr_21 = V121d * 5.  
IF (V122a > 0 OR V122b > 0) ill_beschr_22 = V122d * 5.  
IF (V123a > 0 OR V123b > 0) ill_beschr_23 = V123d * 5.  
IF (V124a > 0 OR V124b > 0) ill_beschr_24 = V124d * 5.
```

```
COMPUTE ill_beschr = MEAN(ill_beschr_20, ill_beschr_21, ill_beschr_22, ill_beschr_23,  
ill_beschr_24).
```

Zeitpunkt des Ereignisses eruierbar

```
DO IF (MISSING(V10) = 0).  
COMPUTE zeitp_ereign = 10.  
ELSE IF (MISSING(V10) = 1) AND (V11 = 1).  
COMPUTE zeitp_ereign = 5.  
ELSE.  
COMPUTE zeitp_ereign = 0.  
END IF.
```

Impressum vorhanden (Angebotsebene)

```
RECODE V2 (3, 4, 5, 7, 9 = 0) (6, 8 = 3.3333) (1, 2 = 6.6667)  
(10, 11, 12, 13 = 10) INTO impressum.
```

Berechnung des Index für Transparenz

```
COMPUTE transparenz = MEAN(autor_ers, zeitpkt, quel_ers, links_trans,  
method, quell_kont, korrig, korr_bez, meing_tren, ill_beschr,  
zeitp_ereign, impressum).
```

Objektivität

Anteil Meinung am Artikel

```
COMPUTE meing_anteil = 10 - (V52 / 10).
```

```
DO IF V50 = 0.  
COMPUTE meinung = 10.  
ELSE IF V50 = 1 AND V51 = 1.
```

```
COMPUTE meinung = 0.9 * meing_anteil.  
ELSE IF V50 = 1 AND V51 = 0.  
COMPUTE meinung = 0.5 * meing_anteil.  
END IF.  
IF (V53 = 0) meinung = meinung * 0.5.
```

Polemik/Ironie/Sarkasmus

```
COMPUTE iro_pol = MEAN((10 - V54 * 10), (10 - V55 * 10)).
```

Unterschiedliche Positionen

```
DO IF V61 <= 1.  
COMPUTE positionen = 0.  
ELSE IF V61 > 1.  
COMPUTE positionen = V61 * 2.  
ELSE IF V61 > 5.  
COMPUTE positionen = 10.  
END IF.
```

Diskursivität

```
COMPUTE diskurs = V62 * 10.
```

Anzahl Akteure

```
COMPUTE akteure = V63 * 2.5.  
IF V63 > 4 akteure = 10.
```

Veröffentlichung von Kommentaren

```
DO IF V101 = 1.  
COMPUTE kommentare_ver = V102 * 10.  
ELSE IF V101 = 0.  
COMPUTE kommentare_ver = 0.  
END IF.
```

Berechnung des Index für Objektivität

```
COMPUTE objektivitaet = MEAN (kommentare_ver, akteure, diskurs, positionen,  
iro_pol, meinung).
```

Vielfalt

Vielfalt der Verlinkung

```
DO IF V20 > 0.  
COMPUTE links_ext = (10 / V20) * V21.  
COMPUTE links_vielf = MEAN(links_ext, V21).  
END IF.
```

Quellenvielfalt

```
COMPUTE quellen_typen = 0.  
IF (V25 > 0) quellen_typen = quellen_typen + 1.  
IF (V26 > 0) quellen_typen = quellen_typen + 1.  
IF (V27 > 0) quellen_typen = quellen_typen + 1.  
IF (V28 > 0) quellen_typen = quellen_typen + 1.  
IF (V29 > 0) quellen_typen = quellen_typen + 1.  
IF (V31 > 0) quellen_typen = quellen_typen + 1.  
IF (V32 > 0) quellen_typen = quellen_typen + 1.  
IF (V34 > 0) quellen_typen = quellen_typen + 1.  
COMPUTE quellen_typen = quellen_typen * (10/8).  
  
COMPUTE quellen_vielf = MEAN(quellen_typen, (quel_anz * (10/15))).
```

Meinungsvielfalt

```
COMPUTE meing_anz = (V50 + V51).  
RECODE meing_anz (1 = 0) (2 = 10).  
COMPUTE pos_anz = V61 * (10 / 7).  
COMPUTE vielf_meing = MEAN(meing_anz, pos_anz).
```

Vielfalt der Qualitätskriterien

```
COMPUTE anz_kriterien = (V85 + V86 + V87 + V88 + V89 + V90 + V91 + V92  
+ V93 + V94 + V95 + V96 + V97 + V98 + V99) * (10/15).  
IF V80 = 0 anz_kriterien = 0.
```

Vielfalt der Illustrationen

```
COMPUTE ill_anz = V120a + V120b + V121a + V121b + V122a + V122b + V123a  
+ V123b + V124a + V124b.  
COMPUTE ill_anz = ill_anz * (10 / 6).  
IF ill_anz > 10 ill_anz = 10.  
  
COMPUTE anz_ill_typ = 0.  
IF (V120a > 0 OR V120b > 0) anz_ill_typ = anz_ill_typ + 1.  
IF (V121a > 0 OR V121b > 0) anz_ill_typ = anz_ill_typ + 1.  
IF (V122a > 0 OR V122b > 0) anz_ill_typ = anz_ill_typ + 1.  
IF (V123a > 0 OR V123b > 0) anz_ill_typ = anz_ill_typ + 1.  
IF (V124a > 0 OR V124b > 0) anz_ill_typ = anz_ill_typ + 1.  
COMPUTE anz_ill_typ = anz_ill_typ * 2.  
  
COMPUTE vielf_ill = MEAN(ill_anz, anz_ill_typ).
```

Themenvielfalt (Angebotsebene)

```
RECODE V2 (1, 9 = 6) (2, 4 = 9) (3, 12 = 5) (5, 6, 7, 8, 10 = 7)  
(11 = 3) (13 = 1) INTO vielf_themen.  
COMPUTE vielf_themen = vielf_themen * (10 / 11).
```

Vielfalt der Artikeltypen (Angebotsebene)

```
RECODE V2 (3 = 1) (11 = 2) (1, 2, 5, 7, 8, 9, 10, 13 = 3)  
(4, 6, 12 = 4) INTO vielf_arttyp.  
COMPUTE vielf_arttyp = vielf_arttyp * (10 / 4).
```

Berechnung des Index für Vielfalt

```
COMPUTE vielfalt = MEAN (vielf_ill, anz_kriterien, vielf_meing, quellen_vielf,  
links_vielf, vielf_themen, vielf_arttyp).
```

Vollständigkeit

Textlänge

```
COMPUTE textlaenge = V111 * (10 / 8405).  
IF textlaenge > 10 textlaenge = 10.
```

Berechnung des Index für Vollständigkeit

```
COMPUTE vollstaendigkeit = MEAN(textlaenge, vielf_themen).
```

Relevanz

Themenrelevanz

```
RECODE V60 (1, 2, 3, 8 = 3) (5, 10, 11, 12 = 2) (4, 6 = 1) (7, 9 = SYSMIS)  
INTO rel_thema.
```

```
COMPUTE rel_thema = rel_thema * (10 / 3).
```

Relevanz der Illustrationen

```
COMPUTE anz_ill_typ = 0.  
IF (V120a > 0 OR V120b > 0) anz_ill_typ = anz_ill_typ + 1.  
IF (V121a > 0 OR V121b > 0) anz_ill_typ = anz_ill_typ + 1.  
IF (V122a > 0 OR V122b > 0) anz_ill_typ = anz_ill_typ + 1.  
IF (V123a > 0 OR V123b > 0) anz_ill_typ = anz_ill_typ + 1.  
IF (V124a > 0 OR V124b > 0) anz_ill_typ = anz_ill_typ + 1.
```

```
RECODE V120c (SYSMIS = 0).  
RECODE V121c (SYSMIS = 0).  
RECODE V122c (SYSMIS = 0).  
RECODE V123c (SYSMIS = 0).  
RECODE V124c (SYSMIS = 0).
```

```
COMPUTE ill_bezug = MAX(V120c, V121c, V122c, V123c, V124c) * 5.  
DO IF anz_ill_typ = 0.  
RECODE ill_bezug (0 = SYSMIS).  
END IF.
```

Berechnung des Index für Relevanz

```
COMPUTE relevanz = MEAN(ill_bezug, rel_thema, aktualitaet).
```

Analytische Qualität

Interne Einordnung (Interne Verlinkung, Bezugnahme auf ältere Artikel)

```
COMPUTE links_dummy = V20 - V21.  
IF V20 = 0 links_int = 0.
```

```
IF links_dummy > 0 links_intern = 5.  
IF links_dummy >= 5 links_intern = 10.  
COMPUTE bez_int = MEAN((V13 * 10), links_intern).
```

Kritik

```
COMPUTE kritik = MEAN((V80 * 10), anz_kriterien).
```

Berechnung des Index für analytische Qualität

```
COMPUTE analytische_qualitaet = MEAN(bez_int, meing_tren, diskurs, kritik).
```

Interaktivität

Kommentare

```
COMPUTE kommentare = MEAN(V101, kommentare_ver, beantw).
```

Interaktionsmöglichkeiten auf Angebotsebene

```
RECODE V2 (3, 7, 8, 11, 12, 13 = 1.6667) (1, 2, 4, 5, 6, 9, 10 = 3.3333)  
INTO int_angeb.
```

Berechnung des Index für Interaktivität

```
COMPUTE interaktivitaet = MEAN((V33 * 10), (V100 * 5), kommentare, int_angeb).
```

Verständlichkeit

Berechnung des Index für Verständlichkeit

```
COMPUTE verst = (V110 * (100 / 80)).  
IF verst > 100 verst = 100.
```

```
COMPUTE verstaendlichkeit = verst / 10.
```

Hypermedialität

Verlinkung

```
COMPUTE verlink_dummy = V20.  
IF V20 > 10 verlink_dummy = 10.  
COMPUTE verlinkung = MEAN(verlink_dummy, links_deep).
```

Onlinespezifische Interaktionsmöglichkeiten

COMPUTE hyper_inter = MEAN((V100 * 5), (V101 * 10)).

Werbung

COMPUTE werbung = MEAN((10 - (V125 * 10)), (10 - (V127 / 10)), (V128 * 10)).

Hypermedialität auf Angebotsebene

RECODE V2

(1 = 7.63) (2 = 6.43) (3 = 7.5) (4 = 6.82) (5 = 6.73) (6 = 7.6)
(7 = 6.67) (8 = 8.38) (9 = 7.63) (10 = 6.73) (11 = 1.67)
(12 = 5.83) (13 = 5.83)

INTO hyper_angeb.

Berechnung des Index für Hypermedialität

COMPUTE hypermedialitaet = MEAN(verlinkung, hyper_inter, vielf_ill,
werbung, hyper_angeb).

Unterhaltsamkeit

Berechnung des Index für Unterhaltsamkeit

COMPUTE unterhaltsamkeit = MEAN(vielf_ill, (V54 * 10), (V55 * 10)).

Gesamtwerte

Berechnung der Gesamtwerte

COMPUTE gesamtnote = MEAN(aktualitaet, transparenz, objektivitaet, vielfalt,
relevanz, analytische_qualitaet, interaktivitaet, verstaendlichkeit,
hypermedialitaet, vollstaendigkeit, unterhaltsamkeit).

Typenbildung

RECODE V2 (1=2) (2=2) (3=1) (4=2) (5=1) (6=1) (7=2) (8=1) (9=2) (10=2) INTO gruppe_einzel.
RECODE V2 (1=1) (2=1) (3=2) (4=2) (5=2) (6=1) (7=2) (8=1) (9=2) (10=2) INTO anonym_unan.

Recodierung auf Angebotsebene

Publikationshäufigkeit

COMPUTE publ_haeuf = V210 / 6.233.

Impressum

```
COMPUTE impressum = V236 * (10 / 3).
```

Interaktivität

```
COMPUTE int_angeb = MEAN(V220, V221, V222, V223, V224, V225, V226) * 10.
```

Hypermedialität

```
RECODE V233 (1, 2, 3 = 10) (9 = SYSMIS) INTO spambots.
```

```
RECODE V234 (0 = 10) (1 = 0) (2 = 5) INTO email_sicht.
```

```
COMPUTE zugang = MEAN((10 - (V230 * 10)), (10 - (V231 * 10)), (V232 * 10)).
```

```
COMPUTE hyper_angeb = MEAN(zugang, spambots, email_sicht, (V235 / 10)).
```